Program WOOF: A Numerical Evaluator of Loudspeaker Systems

MICHAEL L. LAMPTON

Abstract—A Fortran language computer program is described that evaluates the absolute low-frequency response and electrical impedance functions of frequency for direct radiator electrodynamic loudspeakers mounted in vented, unvented, or infinite baffle enclosures. The program is intended to be used interactively, with the engineer serving to suggest modifications of a design and the program providing the modified response plots. The program listing, typical input data, and an example of the program's output are shown.

Introduction

Although mathematical expressions are available that describe the absolute low-frequency response of direct radiator loudspeaker systems (see, for example, [1, eq. 15]), a systematic, accurate, and rapid method of evaluating these expressions has been lacking. This situation prompted the author to develop a computer program that evaluates and plots the response and impedance versus frequency curves for a given design. The program presents, in addition, a variety of tabular data that will often enable the experienced engineer to improve existing designs. Finally, the program minimizes the use of central processor time and input–output facilities, and can be satisfactorily used by means of remote terminal Teletype stations.

Theory

The MKS mechanical impedance z_1 of a piston of area A_1 and mass m_1 radiating on one side into a semi-infinite space, and enclosed on the other side by a vented box of volume V, is [1]–[3]

$$z_1 = r_{\text{rad}} + r_1 + i\omega m_1 + \frac{s_1}{i\omega} + F\frac{s_2}{i\omega} \tag{1}$$

where r_{rad} represents the radiation resistance given by $0.022f^2A_1^2$ MKS·Ω for wavelengths substantially greater than the piston diameter; r_1 is the resistance of the suspension; s_1 represents the suspension stiffness; and s_2 represents the enclosure air stiffness given by $\gamma p_0 A_1^2/V$ with V the adiabatic enclosure volume and with $\gamma p_0 = 143\ 000\ \text{n/m}^2$. F is the complex ratio of the vent plus piston volume velocity to the piston volume velocity, and can be calculated from elementary circuit theory (see, for example, [1, fig. 4]):

$$F = \frac{r_2 + i\omega m_2}{r_2 + i\omega m_2 + s_2/i\omega}. \tag{2}$$

Here, r_2 and m_2 represent the resistance and mass of air in the vent, respectively. In the present study, r_2 is taken to equal r_{rad}. Sealed boxes have $m_2 = \infty$ and $F=1$. Infinite baffles have $V=\infty$, $s_2=0$, and $F=1$.

The absolute response at each frequency is given by the ratio of radiated acoustic power to the mean-square voice coil voltage divided by the electrical voice coil resistance R. A derivation of the absolute response appears, for example, in Lea and Lampton [3]:

$$E = \frac{r_{em} r_{\text{rad}} |F|^2}{|z_1 + r_{em}|^2}. \tag{3}$$

In this expression, r_{em} denotes the electrodynamic drag $(BL)^2/R$. E is a function of frequency because r_{rad}, F, and especially z_1 are frequency dependent.

A useful diagnostic of a system's performance is its electrical impedance curve, which essentially reflects the frequency dependence of z_1:

$$Z_{el} = R\left(1 + \frac{r_{em}}{z_1}\right). \tag{4}$$

This is the sum of the electrical resistance and the motional impedance described by Roder [4] or Olson [2]. A plot of this function of frequency is particularly useful in designing bass reflex systems, since its maxima occur at the lower and upper critical frequencies and its minimum occurs at the enclosure's Helmholtz frequency. When comparing existing systems with theoretical performance predictions, the electrical impedance curve is far easier to measure than the response, and shows any discrepancies in tuning equally well.

Structure

A complete Fortran listing of the program is given in Fig. 1. The sequence of logical processes is most conveniently understood in terms of the six structural divisions of the program. Briefly, these are: 1) declaration of the required variables and arrays, 2) data input and output check, 3) calculation and listing of constant parameters, 4) computation of the response and impedance functions, 5) filling the graphical display matrix, and 6) printing the display matrix. At the end of the listing in Fig. 1 there appears a correctly formatted one-line data input statement.

Manuscript received May 1, 1972.
The author is with the Space Sciences Laboratory, University of California, Berkeley, Calif. 94720.

```
      PROGRAM WOOF(INPUT,OUTPUT)
      DIMENSION RESP(49),ZEL(49),JY(4),JZ(4),J(6),MATRIX(49,29),MK(3)
      REAL M,M1,M2,M2OPT
      COMPLEX F,Z1
      DATA MK(1),MK(2),MK(3)/1H ,1H+,1H./
      DATA JY/-10,-20,-30,-40/
      DATA JZ/150,100,50,0/
      DATA  J/0,20,40,60,80,100/
COMPLETED DECLARATION OF VARIABLES.
    6 READ1,A1,V,FAR,S1,BL,R,Q0,M2
    1 FORMAT(8F8.4)
      PRINT 10
   10 FORMAT(///18X,26HWOOFER PERFORMANCE PLOTTER //)
      PRINT20,A1,V,FAR,S1,BL,R,Q0,M2
   20 FORMAT(4X,24HSYSTEM INPUT DATA ARE... /
     228H EFFECTIVE PISTON AREA       ,F6.3,17H SQUARE METERS    /
     328H ADIABATIC ENCLOSURE VOLUME ,F6.3,17H CUBIC METERS     /
     428H WOOFER FREE AIR RESONANCE  ,F6.1,17H HERTZ            /
     528H SUSPENSION STIFFNESS        ,F6.0,17H NEWTON/METER     /
     628H B L PRODUCT                 ,F6.1,17H WEBER/METER      /
     728H VOICE COIL RESISTANCE       ,F6.1,17H OHMS             /
     828H SUSPENSION Q FACTOR         ,F6.2,17H                  /
     928H REFLEX VENT AIR MASS        ,F6.3,17H KILOGRAM         //)
CONCLUDES INPUT AND OUTPUT ECHO CHECK.
COMMENCE EVALUATING CONSTANT PARAMETERS.
      M1=S1/(6.283*FAR)**2
      S2=143000.*A1**2/V
      M2OPT=M1*S2/S1
      REM=BL**2/R
      FH=0.159*SQRT(S2/M2)
      ASYMP=.0552*REM*(A1/M1)**2
      X1=SQRT(S1*M1)
      R1=X1/Q0
      Q=X1/(R1+REM)
      S=S2/S1
      M=M2/M1
      A=Q**(-2.)-2.-2.*S-2.*S/M
      B=1.+2.*S+S*S+4.*S/M+2.*S*S/M+S*S/M/M-2.*S/Q/Q/M
      C=S*S/Q/Q/M/M-2.*S/M-2.*S*S/M-2.*S*S/M/M
      D=S*S/M/M
      PRINT30,M1,S2,REM,FH,M2OPT,ASYMP,Q,S,M,A,B,C,D
   30 FORMAT(4X,43HFROM THESE DATA, CONSTANT PARAMETERS ARE... /
     228H EFFECTIVE WOOFER MASS       ,F6.3,17H KILOGRAM         /
     328H ENCLOSURE AIR STIFFNESS     ,F6.0,17H NEWTON/METER     /
     428H ELECTRODYNAMIC DRAG         ,F6.2,17H NEWTON SEC/METER/
     528H HELMHOLTZ FREQUENCY         ,F6.1,17H HERTZ            /
     628H SUGGESTED VENT AIR MASS     ,F6.3,17H KILOGRAM         /
     728H ASYMPTOTIC EFFICIENCY       ,F6.3,17H PERCENT          /
     8 6H    Q=,F6.3,6H    S=,F6.3,6H    M=,F6.3/
     9 6H    A=,F6.2,6H    B=,F6.2,6H    C=,F6.2,6H    D=,F6.2//)
COMMENCE EVALUATING RESPONSE AND IMPEDANCE.
      DO100 I=1,49
      FR=2*I
      W=6.283*FR
      RRAD=.022*(FR*A1)**2
      R2=RRAD
      F=CMPLX(R2,W*M2)/CMPLX(R2,W*M2-S2/W)
      Z1=CMPLX(R1+RRAD, W*M1-S1/W)-(0.,1.)*F*S2/W
      RESP(I)=RRAD*REM*CABS(F)**2/CABS(Z1+REM)**2
  100 ZEL(I)=R*CABS(1.+REM/Z1)
CLEAR AND FILL DISPLAY MATRIX.
      DO200 IFREQ=1,49
      DO150 LEVEL=1,29
  150 MATRIX(IFREQ,LEVEL)=MK(1)
      IY=40.5+10.*ALOG10(RESP(IFREQ))
      IF((IY.LT.1).OR.(IY.GT.29))170,162
  162 MATRIX(IFREQ,IY)=MK(2)
  170 IZ=0.5+0.2*ZEL(IFREQ)
      IF((IZ.LT.1).OR.(IZ.GT.29))200,172
  172 MATRIX(IFREQ,IZ)=MK(3)
  200 CONTINUE
COMPLETE MATRIX IS NOW STORED. NEXT, PRINT IT.
      PRINT 600
  600 FORMAT(9X,51HABSOLUTE RESPONSE IN DB(+) AND IMPEDANCE IN OHMS(.))
      PRINT601
  601 FORMAT(6(9X,1H+))
      PRINT602,JY(1),JZ(1)
  602 FORMAT(I9,51(1H+),I3)
      DO 670 LOSS=1,29
      IY=30-LOSS
      IF(LOSS-10) 615,616,615
  615 IF(LOSS-20) 617,618,617
  617 PRINT620,(MATRIX(IFREQ,IY),IFREQ=1,49)
  620 FORMAT(9X,1H+,49A1,1H+)
      GO TO 670
  616 PRINT630,JY(2),(MATRIX(IFREQ,IY),IFREQ=1,49),JZ(2)
  630 FORMAT(I9,1H+,49A1,1H+,I3)
      GO TO 670
  618 PRINT630,JY(3),(MATRIX(IFREQ,IY),IFREQ=1,49),JZ(3)
  670 CONTINUE
      PRINT602,JY(4),JZ(4)
      PRINT601
      PRINT680,J
  680 FORMAT(6(7X,I3))
      PRINT690
  690 FORMAT(25X,17HFREQUENCY,  HERTZ  ///)
      STOP
      END

 .032    .12943  30.0    800.    8.7982  8.00    3.00    .031844
```

Fig. 1. Fortran listing for program WOOF. Following the program there appears a typical data input statement.

The program's name, in and out file requirements, and variable and array names are declared in the first four statements. The following DATA statements fill those arrays that are to be used to label the tick marks on the plotted graph.

The list of input data is read by the program in the statement numbered 6. Care should be exercised that all data are in the MKS system and are properly formatted. As presently written, the input format is 8F8, i.e. exactly eight data of eight characters each, including a decimal point. The required data are: piston area, enclosure volume, woofer free air resonance, woofer suspension stiffness, BL product, total resistance in the voice coil circuit, open-circuit suspension Q factor, and the enclosure's effective vent air mass. As employed by the program, the enclosure volume is assumed to be filled with adiabatically compressible air; if the box is filled with packing material, the compression will be more nearly isothermal and V should be increased by a factor of nearly 1.4 over its physical value. The vent air mass can be obtained from the vent area A_2 and duct length l from the relation

$$m_2 = 1.2A_1^2(l + 0.96\sqrt{A_2})/A_2 \text{ kg}. \tag{5}$$

Sealed "air suspension" systems are treated by setting $m_2 = 9999$ kg. Infinite baffle systems are treated by placing, in addition, $V = 9999$ m^3. As an aid in verifying that the data are correctly read, and to keep the input and output information together, input data are reproduced in the output record.

A number of useful diagnostic parameters are then calculated. The notation follows standard Fortran arithmetic conventions. The Helmholtz frequency of the enclosure is displayed, and a vent mass is suggested that would bring the Helmholtz frequency into equality with the woofer free air resonance. The dimensionless damping and tuning ratios Q, S, and M (see [3]) are calculated, as well as the coefficients A, B, C, and D of the frequency response polynomial (again, see [3]).

The fourth division of the program is a DO loop extending through statement 100. Here the driving frequency FR steps in increments of 2 Hz from 2–98 Hz. Other ranges can be accommodated, but the display software then also should be modified. At each frequency, the quantities in (1)–(4) are evaluated. The response and impedance values are stored in arrays, and constitute the principal results of the program. Note that the response computation follows from directly evaluating (3) rather than from [3, eq. (3)]. This fact permits WOOF to be used in studies of effects not treated by Lea and Lampton, such as radiation damping of the piston and vent meshes.

The graphical output of the program is prepared by the DO loop which extends through statement 200. In this loop, the contents of the array MATRIX are blanked and then set to "+" at the correct decibel level, and to "." at the correct impedance level at each frequency. Some of the statements in this loop prohibit addressing MATRIX beyond its specified field.

The 600-series statements print the graphical data from top to bottom, including titles, borders, tick marks, tick mark labels, and axis labels. In order that

```
                    WOOFER PERFORMANCE PLOTTER

   SYSTEM INPUT DATA ARE...
EFFECTIVE PISTON AREA          .032 SQUARE METERS
ADIABATIC ENCLOSURE VOLUME     .129 CUBIC METERS
WOOFER FREE AIR RESONANCE      30.0 HERTZ
SUSPENSION STIFFNESS           800. NEWTON/METER
B L PRODUCT                     8.8 WEBER/METER
VOICE COIL RESISTANCE           8.0 OHMS
SUSPENSION Q FACTOR            3.00
REFLEX VENT AIR MASS           .032 KILOGRAM

   FROM THESE DATA, CONSTANT PARAMETERS ARE...
EFFECTIVE WOOFER MASS          .023 KILOGRAM
ENCLOSURE AIR STIFFNESS       1131. NEWTON/METER
ELECTRODYNAMIC DRAG            9.68 NEWTON SEC/METER
HELMHOLTZ FREQUENCY            30.0 HERTZ
SUGGESTED VENT AIR MASS        .032 KILOGRAM
ASYMPTOTIC EFFICIENCY         1.079 PERCENT
   Q=  .383     S= 1.414     M= 1.414
   A=  -.00     B=  -.00     C=   .00      D=  1.00
```



Fig. 2. Example of the output prepared by WOOF.

the graph will appear right side up, the graphical data are printed row after row at right angles to the sequence in which they were calculated (column by column).

An example of the output prepared by the program is shown in Fig. 2. These particular data describe a maximally flat bass reflex design having an asymptotic efficiency of 1.08 percent.

Using WOOF

The utility of WOOF in the optimization of loudspeaker systems is best appreciated when rapid job submittal, execution, and evaluation of results is possible. Under these circumstances, the engineer can obtain perfectly reproducible, highly reliable evaluations of loudspeaker systems, and can learn the consequences of varying a large set of design parameters, all at low cost. When, for example, WOOF is compiled and executed on the CDC 6400 computer, only about 1 s of central processor time is required; at a nominal rate of $400/h, the cost per system evaluation is the order of $0.10.

Acknowledgment

The author wishes to acknowledge many helpful discussions with Dr. L. M. Chase of the Space Sciences Laboratory and with S. Lea of the Department of Astronomy, University of California, Berkeley.

References

[1] J. F. Novak, "Performance of enclosures for low-resonance high-compliance loudspeakers," *IRE Trans. Audio*, vol. AU-7, pp. 5–13, Jan.–Feb. 1959.
[2] H. F. Olson, *Acoustical Engineering*. Princeton, N. J.: Van Nostrand, 1957.
[3] S. M. Lea and M. Lampton, "The theory of maximally flat loudspeaker systems," *IEEE Trans. Audio Electroacoust.*, vol. AU-20, pp. 200–203, Aug. 1972.
[4] H. Roder, "The theory of the loudspeaker," *Radio Eng.*, vol. 16, no. 7, p. 10; no. 8, p. 21; no. 9, p. 24; no. 10, p. 19, 1936.